

Chapter 1

Calling Foreign Code with Spock

Interacting with the external world is a mandatory element of modern systems. Been able to call libraries is often a key success to business. In this chapter, we will explain you how to do that in Pharo using Foreign Function Interface.

Suppose you want to use OpenGL in Pharo. The solution is to make a *binding* of each C library function to a Smalltalk method, and then to call these methods from Smalltalk code. But we should answer several questions: how does a binding call a C function? How do you write such binding? How do we pass heap allocated objects? The libraries and mechanisms in charge of doing this kind of work are usually called FFI, for Foreign Function Interface. In this chapter we will help you to use this interface by exposing some practical examples of how to solve common problems faced when communicating with other languages.

Just to mention, there is another possible approach that would be to invoke C libraries by defining a specific plugin for the virtual machine, but it's more complex and it offers no important advantage compared to the results of using bindings in the same language and with the same constructions you are used to.

We will start to show you how to use Spock , a new foreign function library. Spock is based on NativeBoost and previous FFI libraries. Luc ► *Alien and FFI. cite them here no?* ◀

1.1 Getting Started

Luc ► Before the section “calling Ext func”, I think we should first introduce the basics (part of the old “What is a FFI and how it works?” section). Because the reader will want to test while reading. And for that he probably will need a special VM (NativeBoost VM), an image with NativeBoost inside, ... ◀

Right now to run Spock, the new FFI library described in this chapter, you will need a vm that has the NativeBoost plugin and you can find one at <https://ci.lille.inria.fr/pharo/job/NB-Cog-Mac-Carbon/>.

Then you should load in your image, the package NativeBoost-Installer that you can load as follows:

```
Gofer it
  squeaksource: 'NativeBoost';
  package: 'NativeBoost-Installer';
  load.
```

Once loaded you should execute the following expression: NBInstaller install.

1.2 Calling External Functions

Starting Example

Suppose you want to know the amount of time the image has been running by calling the underlying OS function named clock. This function is part of the standard C library. Its C declaration is¹:

```
int clock (void)
```

To call clock from the image, you will write a binding, which is a normal method with a <primitive:module:> pragma. This pragma describes the signature of the C function. Here is the definition. We explain below the details.

```
CExamples class>>ticksSinceStart
  <primitive: #primitiveNativeCall module: #NativeBoostPlugin>

  ↑ NBFFICallout cdecl: #( int clock () ) module: NativeBoost CLibrary
```

Stéf ► Igor in 1.4 I get a vmVersion -> deprecated. ◀

¹ According to man its return type is clock_t instead of int, but we deliberately made it int for the sake of simplicity. **Stéf** ► how would we deal with that if we do not want/can change clock_t into int? ◀

You can define this method in any class, on its instance or class side. It doesn't make any difference. In the above example, we defined a class named `CExamples` and define the method on its class side. This way it is easy to access it by sending a message to this class without creating unnecessary instances.

Then, to know the time since the program started you can execute and print it. Note that these are just the native clock ticks, which aren't of much use.

Script 1.1: Invoking a C binding

```
CExamples ticksSinceStart
```

Analysis

Now that the call worked, let us look at the definition again:

```
CExamples class>>ticksSinceStart
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ NBFFICallout cdecl: #( int clock () ) module: NativeBoost CLibrary
```

Callout. What we have just done is usually called an FFI *callout*. `NBFFICallout` is the class that defines behavior to perform such calls to external functions. We use the message `#cdecl:module:` to describe this call.

This message has two arguments: the *signature of the function* that is about to be called and the *module or library* where to look for it which is the standard C library in this example.

The signature is described by an array whose first element is the C return type and the second is the name of the called function. The third element is also an array which describes the arguments of the function when there's any.

About conversion. The return type is `int` and not `SmallInteger`. This is the same with function arguments. They all describe in terms of C language. But you don't have to bother with this because the Spock library does an automatic conversion between C types and Smalltalk types. So, when this message is executed, the return value will be a `SmallInteger`. This conversion process for types from different languages is called *marshaling*. We will see more examples of automatic conversions in this chapter.

There are always three things to take into account when doing callouts. First, of course, is the signature of the C function we are going to call, the

definition of the function from a C point of view. Through this book we put in *C header* box. We also have to think on the way we access this function from Smalltalk world, on the code that sends the message to use the bounded C function. We put that in *Library access*. Finally, the complex part, is how we mix these two worlds, the C and the smalltalk worlds. That's the *Pragma declaration*. This Pragma declaration is the code that describes the C header and 'binds' Smalltalk arguments to C arguments. Let's look at a summary of our first example then:

C header.

```
int clock (void)
```

Pragma declaration.

```
CExamples class>>ticksSinceStart
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>
↑ NBFFICallout cdecl: #( int clock () ) module: NativeBoost CLibrary
```

Library access.

 CExamples ticksSinceStart

Passing arguments to the function

The previous example was the simplest we could find: we asked for the execution of a function without parameters and got the results. Now let's look how we can execute functions that require arguments.

Passing an integer

Let's try with a really simple function, *abs*, which takes an integer and returns its absolute value.

C header.

```
int abs ( int n );
```

Pragma declaration.

```
CExamples class>>abs: anInteger
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ NBFFICallout cdecl: #( int abs (int anInteger) ) module: NativeBoost CLibrary
```

Library access.

 CExamples abs: -20

Stef ▶ *here*◀ Can you see the small differences with the previous example? First of all, in this case we are specifying `abs` instead of `clock`, of course. But besides that, the only change is that we added an argument. When functions have arguments you have to specify two things for each of them: its type and the object that you want to have sent to the C function. That is, in the arguments array, we put the type of the argument, and after that, the name of the Smalltalk variable we are going to send as argument. This type-and-name pair will be repeated, separated by comma for each argument, as we will show in the next examples. In this case, the argument is a C integer, so you put `int`. As the object we want to pass is the same argument we received, we directly put its name there in the declaration.

Now you can try printing this:

 CExamples abs: -20.

Before getting into the next example, a little thing needs to be remarked. In the pragma declaration we are expressing not one but *two* different aspects: the obvious one is the C function signature, the other is which objects we are passing as arguments to the C function when the method is invoked to receive the expected result. In this second aspect there are many possibilities. In this case the argument of the C function comes directly as an argument of the method: `anInteger`. But it is not always necessary to take the objects to send as arguments. You can also put constants, `self` or any instance variable as an argument to the C function. Suppose you want to add this `abs` wrapper to `SmallInteger`, so that you can do

 -50 *absoluteValue*

In that case you should add `absoluteValue` method to `SmallInteger`, but instead of passing an argument you directly pass `self`!

```
SmallInteger>>absoluteValue
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ NBFFICallout cdecl: #( int abs (int self) ) module: NativeBoost CLibrary
```

Or in other circumstance you may want to have a wrapper that always calls `abs` function with `-45` value. Then you could directly write

```
CExamples class>>absMinusFortyFive
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ NBFFICallout cdecl: #( int abs (-45) ) module: NativeBoost CLibrary
```

Notice that this time you don't put the type of the argument. It is also possible to pass an instance variable. We let that as a really easy exercise.

Passing strings

Now we will call the very well known `strlen` function. This function requires a string as argument and returns its number of characters.

C header.

```
int strlen ( const char * str );
```

Pragma declaration.

```
CExamples class>>stringLength: aString
  <primitive: #primitiveNativeCall module: #NativeBoostPlugin>

  ↑ NBFFICallout cdecl: #( int strlen (String aString) ) module: NativeBoost
    CLibrary
```

Library access.

 *CExamples stringLength: 'awesome!'*

In the signature department there are two small differences between the C signature and the one we are providing here in the Smalltalk array. The first difference is the `const` keyword of the argument. For those not used to C, that's only a modifier keyword that the compiler take advantage of to make some static validations at compile time. It's has no value when describing the signature for calling a function at runtime. The second difference, an important one, is that as argument we are providing `String aString` instead of `char* aString`. It is important to remark again that Spock will take care of automatically doing the conversion of the arguments, which come from Smalltalk world, to C values. But as this argument is not a common char pointer but a C string, then a null terminator has to be present at its end. To make Spock know this is why you specify the argument as `String` and not as `char*`. *This means that the string passed in the example is going to be put in an external C char array and that a null termination character will be added to it (as Smalltalk strings are not null terminated). Also, this array will be automatically released after the call ends.* Conversely, Spock will take the C result value of calling the C function and convert it to a proper Smalltalk object, an Integer in this particular case.

To continue with the examples we will call `strcmp`, which takes two arguments and returns -1, 0 or 1 depending on the relationship between both strings.

C header.

```
int strcmp ( const char * str1, const char * str2 );
```

Pragma declaration.

```
stringCompare: aString with: anotherString

<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ NBFFICallout cdecl: #( int strcmp (String aString, String anotherString) )
  module: NativeBoost CLibrary
```

Library access.

 CExamples stringCompare: 'awesome!' with: 'awesome'

Notice that you can add arguments by appending them to the arguments array, using a comma to separate them. Also notice that you have to explicitly tell which object is going to be sent for each argument, as already told. In this case, aString is the first one and anotherString is the second.

The next example adds a bit more complexity: we are going to send a ByteArray to the memset function.

C header.

```
void * memset ( void * ptr, int value, int num );
```

Sets the first num bytes of the block of memory pointed by ptr to the specified value

Pragma declaration.

```
memorySet: aByteArray with: anInteger for: aByteCount

<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ NBFFICallout cdecl: #( void* memset (char* aByteArray, int anInteger, int
  aByteCount) ) module: NativeBoost CLibrary
```

Library access.

 array := ByteArray new: 10.
CExamples memorySet: array with: 100 for: 5.
array

As you should notice, this time Spock takes care of finding out the memory address where `aByteArray` lives and sends a pointer to it as the first argument. After executing the example you can see that the byte array elements were actually changed. This is an important difference with strings. The expertise user would notice that string functions in C language expects to receive a pointer to contiguous memory bytes ending with a null character to delimit its length. Spock is responsible for copying the string to a bigger buffer and inserting this special end character. This means that strings are passed by-copy, while on the other hand byte arrays are passed by reference!. Also for expertise users, you may notice that in this last example, as sending a reference, we are sending to a native library a pointer to an object that lives in Smalltalk's memory. We already know that Smalltalk's memory is auto cleaned by a garbage collector that potentially move the objects to other places, so we have to be careful about this special behaviour. At the moment this book is written the FFI library doesn't support asynchronous calls, so strange behaviour is not possible. But we have to be very careful when facing with C pointer arguments to Smalltalk's memory.

Using C structures

Loading code from other libraries

1.3 Wrapping Cairo library

Cairo is a drawing library written in C. Here we present the basis of writing a wrapper for it, which should be complete enough to write any wrapper for any C library.

Small example

Whenever you want to start writing a wrapper, you should do it this way: find the smallest piece of code you can find that uses that library, and wrap everything you need to make it work from Smalltalk. Then you can go on adding more and more to the wrapped functionality.

In this example we have a minimal program that uses Cairo library, taken from Cairo's FAQ [Javier](#) ► *Add url to it* ◀. It creates a Cairo surface, puts some text on it and renders everything to a PNG.

```
#include <cairo.h>

int
main (int argc, char *argv[])
{
```

```

cairo_surface_t *surface =
    cairo_image_surface_create (CAIRO_FORMAT_ARGB32, 240, 80);
cairo_t *cr =
    cairo_create (surface);

cairo_select_font_face (cr, "serif", CAIRO_FONT_SLANT_NORMAL,
    CAIRO_FONT_WEIGHT_BOLD);
cairo_set_font_size (cr, 32.0);
cairo_set_source_rgb (cr, 0.0, 0.0, 1.0);
cairo_move_to (cr, 10.0, 50.0);
cairo_show_text (cr, "Hello, world");

cairo_destroy (cr);
cairo_surface_write_to_png (surface, "hello.png");
cairo_surface_destroy (surface);
return 0;
}

```

The way we will work then is to wrap everything we find in our way from the first to the last line. In order to wrap a library, you should get its reference documentation, so that you know exactly which are its structures, and functions. We start by creating some classes for all Cairo structures that are going to be used here and there in the wrapper.

We create a class for all the library's constants (we will fill them later):

Class 1.2:

```

SharedPool subclass: #NBCairoConstants
instanceVariableNames: "
classVariableNames: "
poolDictionaries: "
category: 'NBCairo-Core'

```

and a class for all the library's typedefs and enums (we will also fill them later):

Class 1.3:

```

SharedPool subclass: #NBCairoTypes
instanceVariableNames: "
classVariableNames: "
poolDictionaries: "
category: 'NBCairo-Core'

```

Notice that both of them are Shared Pools. We now create a base class for structures used in the wrapper, putting the shared pools, so that we can use all of them in the derived classes.

Class 1.4:

```
NBExternalObject subclass: #NBCairoHandle
  instanceVariableNames: "
  classVariableNames: "
  poolDictionaries: 'NBCairoConstants NBCairoTypes'
  category: 'NBCairo-Core'
```

Finally, to wrap the first line of main function we will create a class to represent a Cairo surface.

Class 1.5:

```
NBCairoHandle subclass: #NBCairoSurface
  instanceVariableNames: "
  classVariableNames: "
  poolDictionaries: "
  category: 'NBCairo-Core'
```

With most well written libraries, the contents of the structures they declare is private to their internals. They are opaque, in the sense that you may not know nor care what bits they have inside, because you only change them through calls to the library. For that kind of structures, Spock has the `NBExternalObject`. Our wrapping classes inherit from `NBExternalObject`, which is made for this situation, where don't need to fill the fields of the structs.

Now, `NBCairoSurface` is described as an abstract structure, there are different concrete ones, like the image surface, so we add it as a subclass.

Class 1.6:

```
NBCairoSurface subclass: #NBCairoImageSurface
  instanceVariableNames: "
  classVariableNames: "
  poolDictionaries: "
  category: 'NBCairo-Core'
```

Then we can finally create a factory method, to wrap the `cairo_image_surface_create` function. We go to the class side and we enter:

```
NBCairoImageSurface class>>create: aFormat width: aWidth height: aHeight

<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ self call: #(NBCairoImageSurface cairo_image_surface_create (int aFormat,
                                                             int aWidth,
                                                             int aHeight) )
```

and of course, we shouldn't forget to wrap the code that destroys the instance on the C side, but this time on the instance side of the `NBCairoSurface` class:

```
NBCairoSurface>>destroy
  <primitive: #primitiveNativeCall module: #NativeBoostPlugin>

  ↑ self call: #(void cairo_surface_destroy (NBCairoSurface self) )
```

In the creation method, we declare the function as returning a `NBCairoImageSurface`, but actually it is returning a `cairo_surface_t` pointer. The same happens on the destroy method which is described as an `NBCairoSurface`, but actually is a `cairo_surface_t` pointer. In both cases, spock takes care of the conversion.

These methods are almost working. To finish, we add some helper functions to the `NBCairoHandle` class, the one which was going to contain some useful configuration:

```
NBCairoHandle class>>libraryHandle
  ↑NativeBoost forCurrentPlatform loadModule: '/usr/lib/libcairo.so.2'

NBCairoHandle>>libraryHandle
  ↑self class libraryHandle

NBCairoHandle(both instance and class side)>>call: fnSpec
  " you can override this method if you need to"

  | sender |

  sender := thisContext sender.
  ↑NBFFICallout
    handleFailureIn: sender
    nativeCode: [ :gen |
      gen
        sender: sender;
        cdecl;
        generateCall: fnSpec module: self libraryHandle]
```

Javier ► we should change something in order to remove this last method ◀

Now to be able to test a bit, we have to add some more things. We add `cairo_status_t` (which is an enum) to the `NBCairoTypes`.

Class 1.7:

```
SharedPool subclass: #NBCairoTypes
  instanceVariableNames: "
  classVariableNames: 'cairo_status_t'
  poolDictionaries: "
  category: 'NBCairo-Core'
```

```
NBCairoTypes class>>initialize
```

```
"self initialize"
```

```
cairo_status_t := #int
```

and we doit:

 *NBCairoTypes initialize*

```
.
```

so that `cairo_status_t` gets initialized. The, we add a method to the `NBCairoSurface` class:

```
NBCairoSurface>>status
```

```
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>
```

```
↑ self call: #(cairo_status_t cairo_surface_status (NBCairoSurface self) )
```

We are now ready to test just a bit. Take a second and save the image now. Remember that if you wrongly specify the parameters of an FFI call, you could crash the image.

To test that everything is working do print this in a workspace:

 *surface := NBCairoImageSurface new. surface status*

If everything is fine, status should be 0. Then don't forget to destroy the surface:

 *surface destroy*

To make the example complete, let's add the final pieces needed.

Here is how the original example will look after translating it to use the wrapper:

```
CairoExamples>>minimalProgram
```

```
"
```

```
self new minimalProgram
```

```
"
```

```
| surface context fileName |
```

```
surface := NBCairoImageSurface create: CAIRO_FORMAT_ARGB32 width: 240  
height: 80.
```

```
context := surface createContext.
```

```
context selectFont: 'serif' slant: CAIRO_FONT_SLANT_NORMAL weight:  
CAIRO_FONT_WEIGHT_BOLD;
```

```
setFontSize: 32.0;
```

```
setSourceR: 0.0 G: 0.0 B: 1.0;
```

```
moveToX: 10.0 Y: 50.0;
```

```
showText: 'Hello, world';
destroy.
```

```
fileName := (FileDirectory default / 'Hello.png') fullName.
surface writeToPng: fileName.
surface destroy.
```

```
Form openImageInWindow: fileName
```

but before it works we have to add some constants and types. `CAIRO_FORMAT_ARGB32` is a constant defined on `cairo_format_t`, and we are using constants for font slants and font weights too. For each enum we add a type on `NBCairoTypes` and its constants to `NBCairoConstants`:

```
instanceVariableNames: "
classVariableNames: 'CAIRO_FONT_SLANT_ITALIC
CAIRO_FONT_SLANT_NORMAL CAIRO_FONT_SLANT_OBLIQUE
CAIRO_FONT_WEIGHT_BOLD CAIRO_FONT_WEIGHT_NORMAL
CAIRO_FORMAT_A1 CAIRO_FORMAT_A8 CAIRO_FORMAT_ARGB32
CAIRO_FORMAT_INVALID CAIRO_FORMAT_RGB16_565
CAIRO_FORMAT_RGB24'
poolDictionaries: "
category: 'NBCairo-Core'
```

```
SharedPool subclass: #NBCairoTypes
```

```
instanceVariableNames: "
classVariableNames: 'cairo_font_slant_t cairo_font_weight_t cairo_status_t'
poolDictionaries: "
category: 'NBCairo-Core'
```

and then we add the code to initialize these variables:

```
NBCairoTypes class>>initialize
```

```
"self initialize"
```

```
cairo_status_t := cairo_font_slant_t := cairo_font_weight_t := #int
```

```
NBCairoConstants class>>initialize
```

```
"enum cairo_format_t"
```

```
CAIRO_FORMAT_INVALID := -1.
CAIRO_FORMAT_ARGB32 := 0.
CAIRO_FORMAT_RGB24 := 1.
CAIRO_FORMAT_A8 := 2.
CAIRO_FORMAT_A1 := 3.
CAIRO_FORMAT_RGB16_565 := 4
```

```
"enum cairo_font_slant_t"
CAIRO_FONT_SLANT_NORMAL := 0.
CAIRO_FONT_SLANT_ITALIC := 1.
CAIRO_FONT_SLANT_OBLIQUE := 2

"enum _cairo_font_weight_t"
CAIRO_FONT_WEIGHT_NORMAL := 0.
CAIRO_FONT_WEIGHT_BOLD := 1.
```

don't forget to  *NBCairoConstants initialize*

and  *NBCairoTypes initialize*

.

As a last step, you need to wrap the remaining functions. We leave here a last example and the rest of them you can take the code from the NBCairo repository on squeaksource.

Class 1.8:

```
NBCairoSurface subclass: #NBCairoImageSurface
instanceVariableNames: "
classVariableNames: "
poolDictionaries: "
category: 'NBCairo-Core'
```

```
CairoContext>>selectFont: aFamily slant: slant weight: aWeight
<primitive: #primitiveNativeCall module: #NativeBoostPlugin>

↑ self call: #(void cairo_select_font_face (NBCairoContext self,
String aFamily,
cairo_font_slant_t slant,
cairo_font_weight_t aWeight) )
```

Original FFI stuff

Thanks to a Foreign Function Interface (FFI), it is possible to achieve this and interact with third-party libraries. Multiple FFI libraries are available in Pharo such as FFI (<http://wiki.squeak.org/squeak/1414>) and Alien (<http://www.squeaksource.com/Alien>). As an example the dynamic OpenGL library (.dll, .so or .dylib depending on the operating system) is used in Croquet through the FFI library and AlienOpenGL (<http://www.squeaksource.com/AlienOpenGL>) use it through the Alien library.

In this chapter, we will describe what is a FFI library and how it works with the virtual machine. We will then dive into the Alien library and learn how it can be used thanks to the AlienOpenGL example.

1.4 What is a FFI and how it works?

The Pharo image is loaded and executed by a virtual machine (VM) which is itself executed on the top of the operating system. Two different virtual machines can be used for Pharo images: the squeak VM and the Cog VM. Regarding the general FFI mechanism, it is not really relevant to make a distinction² Both are based on a plugin architecture. A plugin enable the Smalltalk code (in the image) to access third-party functionalities (outside the image and the VM).

The Alien library uses the *IA32ABI* plugin which

1.5 Using the Alien Library

Installation

Alien is a Squeaksource project <http://www.squeaksource.com/Alien>. Use the following code to install Alien in a Pharo 1.1 image:

```
Gofer new
  url: 'http://www.squeaksource.com/Alien' ;
  package: 'ConfigurationOfAlien';
  load.

(Smalltalk at: #ConfigurationOfAlien) loadCore ;
  loadTests ;
  loadLibC.
```

²It worth noting that the Cog VM is younger and really faster than the Squeak VM but it still presents some small limitations and bugs at the time writing this book.




Figure 1.1: Pharo Architecture

Run the tests to ensure that Alien is well installed and that the VM plugin is correctly working.

First example

Core Classes

Figure 1.2 shows an UML diagram of the Alien-Core package that contains the main classes of this library.

Memory management

Calling an external C library from Smalltalk

Callbacks from C to Smalltalk code

1.6 Dissection of Alien

Alien

FFICallbackReturnValue




Figure 1.2: Core classes of the Alien library

FFICallbackThunk

UnsafeAlien

AlienLibrary

Callback

AlienWeakTable

1.7 Study Case: the AlienOpenGL package